# $Id: PKGBUILD 183334 2013-04-21 22:08:33Z heftig $
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=bijiben
pkgver=3.8.1
pkgrel=1
pkgdesc="Write out notes, every detail matters"
arch=(i686 x86_64)
url="https://live.gnome.org/Bijiben"
license=(GPL)
depends=(libtracker-sparql clutter-gtk libzeitgeist webkitgtk3 hicolor-icon-theme desktop-file-utils)
makedepends=(intltool itstool)
options=(!emptydirs)
install=bijiben.install
source=(http://ftp.gnome.org/pub/GNOME/sources/${pkgname}/${pkgver::3}/${pkgname}-${pkgver}.tar.xz)
sha256sums=('a0965874c68ca8492dcffc25cbf6f51549ff4a005c1fa1d87a1d82021aae7908')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var \
    --libexecdir=/usr/lib/$pkgname
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
